
Ex2HR
options nodate nonumber ps=max linesize=75; 
goptions reset=all hsize=5 vsize=4 ftext=simplex;
title;
libname BASUG "d:\my documents\presentations\BASUG_08";

proc print data=BASUG.HR(obs=100);
 var id time wake timebed HR dia hr black pd bmi male age;
run;

*defining dual time variables for 2-part trajectory model (piecewise linear);
data HR; set BASUG.HR;
  if wake = 1 then timewake = time-timebed;
  else timewake = 0;
  if wake = 0 then timesleep = time-timebed;
  else timesleep = 0;
run;

proc print data=HR (obs=100);
 var id time wake timebed timewake timesleep;
run;

*graphical analysis of two-part trajectories;
proc reg data=HR noprint; by id; 
  model HR = timewake timesleep;
  output out=pred p=y_hat;
run;

data pred; set pred;
 if wake = 1 then wakeHR = HR; else wakeHR = .;
 if wake = 0 then sleepHR = HR; else sleepHR = .;
run;

symbol1 value=dot color=black;
symbol2 value=circle color=black;
symbol3 value=none interpol=join color=black h=2 w=2;
axis1 label=("HR") order=(30 to 160 by 10) minor=none;
axis2 label=("Hour") order=(7 to 35 by 7) minor=none;
proc gplot data=pred; by id;
 plot (wakeHR sleepHR y_hat)*time / overlay vaxis=axis1 haxis=axis2 nolegend;
run; quit;

*running two-part model with random slopes for both parts;

*Model 1;
proc mixed data=HR method=reml covtest noclprint;
 class id;
 model HR = timewake timesleep / notest solution ddfm=kr outpred=predicted;
 random intercept timewake timesleep / subject=id type=un gcorr g;
 repeated / type=sp(pow)(time) subject=id rcorr;
run;

symbol1 value=none interpol=join color=black repeat=80;
axis1 label=("HR") order=(40 to 110 by 10) minor=none;
axis2 label=("Hour") order=(5 to 40 by 5) minor=none;
proc gplot data=predicted; where ~missing(HR);
 plot pred*time=id / nolegend vaxis=axis1 haxis=axis2 href=24 lhref=33;
run; quit;

*Model 2;
proc mixed data=HR method=reml noclprint covtest;
 class id;
 model hr = timewake timesleep black timewake*black timesleep*black bmi male age/ 
            notest solution ddfm=kr;

 random intercept timewake timesleep / subject=id type=un gcorr g;
 repeated / type=sp(pow)(time) subject=id;
 ods output solutionf=fixedeffects;
run;

proc print data=fixedeffects; run;
proc transpose data=fixedeffects(keep=estimate effect) out=fixedT; 
 var estimate;
 id effect;
run;
proc print data=fixedT; run;

*generating predicted values for white and black given fixed effect estimates
 holding covariates at female, avg age, avg BMI;
data effects; set fixedT;
 do time = 7 to 36 by 1; 
  if time < 24 then do;  *average time to bed ~ midnight;
     twake = time - 24;
  tsleep = 0;
  end;
  else do;
     twake = 0;
  tsleep = time - 24;
  end;
  whitepred = intercept + timewake*twake + timesleep*tsleep;
  blackpred = whitepred + black + timewake_black*twake + timesleep_black*tsleep;
  output;
 end;
run;

symbol1 value=dot i=join color=black w=1.25 h=1.25 repeat=1;
symbol2 value=square i=join color=green w=1.25 h=1.25 repeat=1;
axis1 label=('HR') minor=none order=40 to 90 by 10;
axis2 label=('Hour') order=5 to 40 by 5 offset=(2) minor=none;
legend1 position=(inside top right) down=3 label=none value=('Black' 'White');
proc gplot data=effects;
 plot (blackpred whitepred)*time /overlay legend=legend1 vaxis=axis1 haxis=axis2 
href=24 lhref=33;
run;

*Model 3;
proc mixed data=HR method=reml noclprint covtest;
 class id;
 model hr = timewake timesleep black timewake*black timesleep*black 
            bmi male age pd timewake*pd timesleep*pd / 
            notest solution ddfm=kr;
 random intercept timewake timesleep / subject=id type=un gcorr g;
 repeated / type=sp(pow)(time) subject=id;
 ods output solutionf=fixedeffects2;
run;

proc transpose data=fixedeffects2(keep=estimate effect) out=fixedT2; 
 var estimate;
 id effect;
run;

*generating predicted values for black and for pd given fixed effect estimates
 holding all other covariates constant;
data effects2; set fixedT2;
 do time = 7 to 36 by 1; 
  if time < 24 then do;  *average time to bed ~ midnight;
     twake = time - 24;
  tsleep = 0;

  end;
  else do;
     twake = 0;
  tsleep = time - 24;
  end;
  whitepred = intercept + timewake*twake + timesleep*tsleep;
  blackpred = whitepred + black + timewake_black*twake + timesleep_black*tsleep;
  * .59 is SD of PD, computing predicted values at +/- 1 SD PD for black 
participants;
  pdhipred = blackpred + pd*.59 + timewake_pd*twake*.59 + timesleep_pd*tsleep*.59;
  pdlopred = blackpred - pd*.59 - timewake_pd*twake*.59 - timesleep_pd*tsleep*.59;
  output;
 end;
run;

symbol1 value=dot i=join color=black w=1.25 h=1.25;
symbol2 value=square i=join color=green w=1.25 h=1.25;
axis1 label=('HR') minor=none order=40 to 90 by 10;
axis2 label=('Hour') order=5 to 40 by 5 offset=(2) minor=none;
legend1 position=(inside top right) down=2 label=none value=('Black' 'White');
legend2 position=(inside top right) down=2 label=none value=('PD = +1SD' 'PD = 
-1SD');
proc gplot data=effects2;
 plot (blackpred whitepred)*time /overlay legend=legend1 vaxis=axis1 haxis=axis2 
href=24 lhref=33;
run;
proc gplot data=effects2;
 plot (pdhipred pdlopred)*time /overlay legend=legend2 vaxis=axis1 haxis=axis2 
href=24 lhref=33;
run;

